

Random Effects Models

Hao Zhang

- Confidence Intervals for Variance Components
 - Confidence Intervals for σ_T^2/σ^2
 - Confidence Intervals for σ_T^2
- Two or More Random Effects
 - Hypothesis Testing:
- Mixed-Effects Models or Mixed Models
 - Confidence Intervals in Mixed Models
- Block Designs and Random Block Effects
- Using SAS
 - Comparing with proc glm
 - When Do proc mixed and proc glm Differ?

$$Y_{it} = \mu + T_i + \epsilon_{it}$$

$$T_i = N(0, \sigma_T^2)$$

$$\epsilon_{it} = N(0, \sigma^2)$$

Confidence Intervals for Variance Components

Confidence Intervals for σ_T^2/σ^2 Represent how strong the random effect is compare to the noise
↳ error term, ϵ_{it}

From (17.3.7), p. 623, we know that

$$\frac{MST}{MSE(c\sigma_T^2/\sigma^2 + 1)} \sim F_{v-1, n-v} \left\{ \begin{array}{l} \text{derived from} \end{array} \right. \rightarrow \begin{array}{l} \frac{MST}{c\sigma_T^2 + \sigma^2} \sim \chi^2_{v-1} \\ \frac{MST}{\sigma^2} \sim \chi^2_{n-v} \end{array}$$

where $c = (n^2 - \sum r_i^2)/(n(v-1))$, and if the r_i are all equal to r , then $c = r$. From this, we can write down an interval in which MST/MSE lies with probability $1 - \alpha$; that is,

$$P\left(F_{v-1, n-v, 1-\alpha/2}^{\text{lower tail}} \leq \frac{MST}{MSE(c\sigma_T^2/\sigma^2 + 1)} \leq F_{v-1, n-v, \alpha/2}^{\text{higher tail}}\right) = 1 - \alpha.$$

If we rearrange the left-hand inequality, we find that

$$c\sigma_T^2/\sigma^2 \leq \frac{MST}{MSE F_{v-1, n-v, 1-\alpha/2}} - 1$$

and similarly for the right-hand inequality,

$$c\sigma_T^2/\sigma^2 \geq \frac{MST}{MSE F_{v-1, n-v, \alpha/2}} - 1.$$

So, replacing the random variables by their observed values, we obtain a $100(1 - \alpha)\%$ confidence interval for σ_T^2/σ^2 as


$$\frac{1}{c} \left[\frac{msT}{msEF_{v-1, n-v, \alpha/2}} - 1 \right] \leq \frac{\sigma_T^2}{\sigma^2} \leq \frac{1}{c} \left[\frac{msT}{msEF_{v-1, n-v, 1-\alpha/2}} - 1 \right].$$

Confidence Intervals for σ_T^2

Note the following unbiased estimator of σ_T^2 ,

$$U = c^{-1}(MST - MSE),$$

where $c = (n^2 - \sum r_i^2)/(n(v-1))$, and $c = r$ when the sample sizes are equal. Let



$$x = \frac{(msT - msE)^2}{msT^2/(v-1) + msE^2/(n-v)}.$$

Then xU/σ_T^2 approximately has the χ^2 distribution with x degrees of freedom. Hence

$$P\left(\chi_{x,1-\alpha/2}^2 \leq \frac{xU}{\sigma_T^2} \leq \chi_{x,\alpha/2}^2\right) \approx 1 - \alpha$$

and

$$P\left(\frac{xU}{\chi_{x,\alpha/2}^2} \leq \sigma_T^2 \leq \frac{xU}{\chi_{x,1-\alpha/2}^2}\right) \approx 1 - \alpha.$$

We have therefore obtained a $(1 - \alpha)100\%$ confidence interval for σ_T^2 .
 $\chi_{0.025}^2$ $\chi_{0.975}^2$ $\alpha = 0.05$
 5% Ex) 95% Confidence.

Two or More Random Effects

In a factorial experiment when both factors have random effects, the two-way complete model is as follows.
 with interaction

$$Y_{ijt} = \mu + A_i + B_j + (AB)_{ij} + \epsilon_{ijt}$$

$$A_i \sim N(0, \sigma_A^2), B_j \sim N(0, \sigma_B^2)$$

$$(AB)_{ij} \sim N(0, \sigma_{AB}^2), \epsilon_{ijt} \sim N(0, \sigma^2)$$

A_i 's, B_j 's, $(AB)_{ij}$'s and ϵ_{ijt} 's are mutually independent

$$t = 1, \dots, r_{ij}, \quad i = 1, \dots, a, j = 1, \dots, b.$$

If σ_{AB}^2 is positive, then there are AB effects present-namely, main effects and interactions for the factors A and B . If σ_A^2 or σ_B^2 is positive, then the corresponding main effects are present.

ANOVA Table:

Source	DF	SS	MS	EMS
A	$a - 1$	SSA	MSA	$br\sigma_A^2 + r\sigma_{AB}^2 + \sigma^2$
B	$b - 1$	SSB	MSB	$ar\sigma_B^2 + r\sigma_{AB}^2 + \sigma^2$
AB	$(a - 1)(b - 1)$	$SSAB$	$MSAB$	$r\sigma_{AB}^2 + \sigma^2$
Error	$ab(r - 1)$	SSE	MSE	

$$\text{F-ratio} = \frac{MSA}{MSAB}, \frac{MSB}{MSAB}, \frac{MSAB}{MSE}$$

Denominator AB :
when factors have random effects.

A Rule for Determining an Expected MS:

First note the subscripts on the term representing the factor in the model. Write down a variance component for the factor of interest, for the error, and for every interaction involving the factor. Multiply each variance components by the corresponding number of observations. Add up terms.

Hypothesis Testing:

$$H_0^{AB} : \sigma_{AB}^2 = 0 \text{ (No interaction) against } H_1^{AB} : \sigma_{AB}^2 > 0.$$

$$\text{Reject } H_0^{AB} \text{ if } F = \frac{MSAB}{MSE} > F_{(a-1)(b-1), n-ab, \alpha}.$$

$$H_0^A : \sigma_A^2 = 0 \text{ against } H_1^A : \sigma_A^2 > 0$$

$$\text{Reject the null hypothesis if } F = \frac{MSA}{MSAB} > F_{a-1, (a-1)(b-1), \alpha}.$$

$H_0^{AB} : \sigma_{AB}^2 = 0$ (Interaction) If variance is 0.
 $H_A^{AB} : \sigma_{AB}^2 > 0$ (No Interaction) If variance is greater than 0.

$\frac{AB}{E}$

$\frac{A}{AB}$

$$H_0^B : \sigma_B^2 = 0 \text{ against } H_1^B : \sigma_B^2 > 0.$$

Reject the null hypothesis if $F = \frac{MSB}{MSAB} > F_{b-1, (a-1)(b-1), \alpha}$.

Note the denominator is not MSE !

Mixed-Effects Models or Mixed Models

Models that contain both random and fixed treatment effects are called mixed models. The analysis of random effects proceeds in exactly the same way as described in the previous sections. All that is needed is a way to write down the expected mean squares. The fixed effects can be analyzed as before, except that, here, too, **we may need to replace the mean square for error by a different one.**

As an example, consider a completely randomized design with three factors A , B and D . We choose a model containing the main effects of factors A , B , and D and the interactions AB and BD . No other interactions are included in the model. Suppose that factors A and B have fixed effects, and factor D has random effects. Then interaction AB is a fixed effect, but interaction BD is a random effect.

- $A+B$ fixed, so AB fixed
- D random, so BD random

Effect	MSE	F - ratio
A	$Q(A, AB) + \sigma^2$	msA/msE
B	$Q(B, AB) + ar\sigma_{BD}^2 + \sigma^2$	$msB/msBD$
D	$abr\sigma_D^2 + ar\sigma_{BD}^2 + \sigma^2$	$msD/msBD$
AB	$Q(AB) + \sigma^2$	$msAB/msE$
BD	$ar\sigma_{BD}^2 + \sigma^2$	$msBD/msE$
Error	σ^2	

For example, the decision rule for testing H_0^B against the alternative hypothesis that the β_j^* are not all equal is

$$\text{reject } H_0^B \text{ if } \frac{msB}{ms(BD)} > F_{(b-1), (b-1)(d-1), \alpha}.$$

To test the hypothesis $H_0^D : \{\sigma_D^2 = 0\}$ against the alternative hypothesis $H_A^D : \{\sigma_D^2 > 0\}$, the decision rule is

$$\text{reject } H_0^D \text{ if } \frac{msD}{ms(BD)} > F_{d-1, (b-1)(d-1), \alpha}.$$

Confidence Intervals in Mixed Models

For fixed effects in a mixed model with equal sample sizes, confidence intervals (including the simultaneous confidence intervals) can be calculated exactly as if there were no random effects in the model, except that we may replace msE used in the denominator of the test ratio by the appropriate one. Apart from this, we may use the Bonferroni, Scheffé, Tukey, and Dunnett methods of multiple comparisons in the usual way.

When the sample sizes are unequal, computing least squares estimates and appropriate standard errors is more complicated. PROC MIXED in SAS software works in this situation.

Block Designs and Random Block Effects

In certain types of experiments, it is extremely common for the levels of a blocking factor to be randomly selected. For example, in medical, psychological, educational, or pharmaceutical experiments, blocks frequently represent subjects that have been selected at random from a large population of similar subjects. In agricultural experiments, blocks may represent different fields selected from a large variable population of fields. In industrial experiments, different machine operators may represent different levels of the blocking factor and may be similar to a random sample from a large population of possible operators. Raw material may be delivered to the factory in batches, a random selection of which are used as blocks in the experiment.

In this case, the block effects will be treated as random effects and all interaction effects with block shall be random as well.

Read the design and analysis of the Temperature Experiment in Section 17.9.

Using SAS

For models with mixed effects, it is recommended to use `proc mixed` over `proc glm`. Only use the latter for equal sample sizes while the

former can be always used and provides more options. `proc mixed` is based on restricted likelihood estimation, and differs from those in the `proc glm`.

Example The Candle Experiment (described on page 667 in Exercise 5). In this complete block design, each experimenter is a block and the only treatment factor is the candle color with 4 levels. We will run a model with block and block by color interaction as random effects.

```
data candle;
  do person=1 to 4;
    do row=1,2;
      do color=1 to 4;
        do col=1,2;
          input time @@;
          output;
          drop row col;
        end; end; end; end;
  lines;
  989 1032 1044 979 1011 951 974 998
  1077 1019 987 1031 928 1022 1033 1041
  899 912 847 880 899 800 886 859
  911 943 879 830 820 812 901 907
  898 840 840 952 909 790 950 992
  955 1005 961 915 871 905 920 890
  993 957 987 960 864 925 949 973
  1005 982 920 1001 824 790 978 938
;
run;
proc print data=candle;
run;

proc mixed data=candle;
class person color;
model time=color;
random person person*color;
lsmeans color /cl pdiff adjust=Tukey;
run;
```

Partial output:

Covariance Parameter Estimates

Cov Parm	Estimate
person	3049.70
person*color	12.2483
Residual	1708.85

The variance of the block effects is estimated as 3049.70 and the variance of the interaction effects is 1708.85.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
color	3	9	11.44	0.0020

The *p*-value for testing that the factor color has equal effects is 0.0020. Note that the number of degrees of freedom for the denominator is 9, which is $(b - 1) \times (v - 1)$ or the degrees of freedom for the interaction term.

~ Alpha, Lower, Upper

Least Squares Means

Effect	color	Estimate	Standard Error	DF	t Value	Pr >
color	1	963.56	29.5346	9	32.62	< .0001
color	2	938.31	29.5346	9	31.77	< .0001
color	3	882.56	29.5346	9	29.88	< .0001
color	4	949.31	29.5346	9	32.14	< .0001

The table provides the least squares estimates of the treatment means of color and the confidence intervals for the means. Note these are not simultaneous confidence intervals.

Differences of Least Squares Means														
Effect	color	_color	Estimate	Standard Error	DF	t Value	Pr > t	Adjustment	Adj P	Alpha	Lower	Upper	Adj Lower	Adj Upper
color	1	2	25.2500	14.8233	9	1.70	0.1227	Tukey-Kramer	0.3757	0.05	-8.2827	58.7827	-21.0253	71.5253
color	1	3	81.0000	14.8233	9	5.46	0.0004	Tukey-Kramer	0.0018	0.05	47.4673	114.53	34.7247	127.28
color	1	4	14.2500	14.8233	9	0.96	0.3615	Tukey-Kramer	0.7739	0.05	-19.2827	47.7827	-32.0253	60.5253
color	2	3	55.7500	14.8233	9	3.76	0.0045	Tukey-Kramer	0.0192	0.05	22.2173	89.2827	9.4747	102.03
color	2	4	-11.0000	14.8233	9	-0.74	0.4770	Tukey-Kramer	0.8778	0.05	-44.5327	22.5327	-57.2753	35.2753
color	3	4	-66.7500	14.8233	9	-4.50	0.0015	Tukey-Kramer	0.0066	0.05	-100.28	-33.2173	-113.03	-20.4747

Figure 1: SAS Output from proc mixed of the Candle Experiment.

Figure 1 provides confidence intervals for pairwise differences, both individual confidence intervals and the simultaneous confidence intervals using Tukey’s method. For example, the 95% confidence interval for $color_1 - color_2$ is (-8.2827, 58.7827) and for the 95% simultaneous confidence intervals, it becomes (-21.0253, 71.5253).

Comparing with proc glm

I will show you the code for proc glm although prefer proc mixed .

```
proc glm data=candle;
class person color; include all factors in 'model'
model time=person color person*color;
random person person*color/test; * Force proc glm to pick right denominator !!!
lsmeans color /cl pdiff adjust=Tukey E=person*color;
run;
```

Note the test option in the random statement is recommended. This option enables the glm procedure to use the identify the correct denominator for hypothesis testing. Therefore, always use the test option in the random statement! See Figure 2. In addition, the E=person*color is necessary for correct confidence intervals. The results from proc mixed and proc glm are identical for this experiment because the equal sample sizes (Figure 3). However, proc glm would require you specify the correct error term in the hypothesis testing and confidence intervals.

The SAS System					
The GLM Procedure					
Source	Type III Expected Mean Square				
person	Var(Error) + 4 Var(person*color) + 16 Var(person)				
color	Var(Error) + 4 Var(person*color) + Q(color)				
person*color	Var(Error) + 4 Var(person*color)				

The SAS System					
The GLM Procedure					
Tests of Hypotheses for Mixed Model Analysis of Variance					
Dependent Variable: time					
Source	DF	Type III SS	Mean Square	F Value	Pr > F
person	3	151659	50553	28.76	<.0001
color	3	60345	20115	11.44	0.0020
Error	9	15821	1757.847222		
Error: MS(person*color)					
Source	DF	Type III SS	Mean Square	F Value	Pr > F
person*color	9	15821	1757.847222	1.03	0.4315
Error: MS(Error)	48	82025	1708.854167		

Figure 2: SAS Output with the Option Test in the Random Statement.

color	time LSMEAN	95% Confidence Limits	
1	963.562500	939.851310	987.273690
2	938.312500	914.601310	962.023690
3	882.562500	858.851310	906.273690
4	949.312500	925.601310	973.023690

Least Squares Means for Effect color				
i	j	Difference Between Means	Simultaneous 95% Confidence Limits for LSMean(i)-LSMean(j)	
1	2	25.250000	-21.025292	71.525292
1	3	81.000000	34.724708	127.275292
1	4	14.250000	-32.025292	60.525292
2	3	55.750000	9.474708	102.025292
2	4	-11.000000	-57.275292	35.275292
3	4	-66.750000	-113.025292	-20.474708

Figure 3: SAS Output with the Correctly Specified Error in the LSMEANS Statement.

GLM: ANOVA

Mixed: REML

When Do proc mixed and proc glm Differ?

proc mixed provides many options to run this procedure, and does not produce identical results as proc glm. The default method in proc mixed is REML (the restricted maximum likelihood). Other estimation methods included in proc mixed include type1, type2, type3, MIVQUE0 and ML. One can specify the method by using the method option in proc mixed. For example,

```
proc mixed method=type3
```

would use the type3 estimation, which is the method used in proc glm.

In general, different methods in proc mixed may yield different estimates, particularly when sample sizes are unequal.

Let us look at the Candle Experiment where the last two data points are dropped. This results in unequal sample sizes.

```
data candle2;
  do person=1 to 4;
    do row=1,2;
      do color=1 to 4;
        do col=1,2;
          input time @@;
          output;
          drop row col;
        end; end; end; end;
  lines;
  989 1032 1044 979 1011 951 974 998
  1077 1019 987 1031 928 1022 1033 1041
  899 912 847 880 899 800 886 859
  911 943 879 830 820 812 901 907
  898 840 840 952 909 790 950 992
  955 1005 961 915 871 905 920 890
  993 957 987 960 864 925 949 973
  1005 982 920 1001 824 790 . .
;
run;
proc print data=candle2;
run;

proc mixed data=candle2;
class person color;
model time=color;
random person person*color;
lsmeans color /cl pdiff adjust=Tukey;
run;

proc glm data=candle2;
class person color;
model time=person color person*color;
random person person*color/test;
lsmeans color /cl pdiff adjust=Tukey E=person*color;
run;
```

Missing

Partial results from proc mixed are provided in Figure 4

Type 3 Tests of Fixed Effects					
Effect	Num DF	Den DF	F Value	Pr > F	
color	3	9	10.89	0.0024	

Least Squares Means									
Effect	color	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
color	1	963.56	29.6011	9	32.55	<.0001	0.05	896.60	1030.52
color	2	938.31	29.6011	9	31.70	<.0001	0.05	871.35	1005.27
color	3	882.46	29.6011	9	29.82	<.0001	0.05	816.60	948.32

Effect	color	_color	Estimate	Standard Error	DF	t Value	Pr > t	Adjustment	Adj P	Alpha	Lower	Upper	Adj Lower	Adj Upper
color	1	2	25.2500	15.1061	9	1.67	0.1290	Tukey-Kramer	0.3905	0.05	-8.9225	59.4225	-21.9082	72.4082
color	1	3	81.0000	15.1061	9	5.36	0.0005	Tukey-Kramer	0.0021	0.05	46.8275	115.17	33.8418	128.16
color	1	4	14.5508	15.6813	9	0.93	0.3777	Tukey-Kramer	0.7912	0.05	-20.9228	50.0244	-34.4030	63.5046
color	2	3	55.7500	15.1061	9	3.69	0.0050	Tukey-Kramer	0.0214	0.05	21.5775	89.9225	8.5918	102.91
color	2	4	-10.6992	15.6813	9	-0.68	0.5122	Tukey-Kramer	0.9013	0.05	-46.1728	24.7744	-59.6530	38.2546
color	3	4	-66.4492	15.6813	9	-4.24	0.0022	Tukey-Kramer	0.0096	0.05	-101.92	-30.9756	-115.40	-17.4954

Figure 4: Proc Mixed Results

Partial results from `proc glm` are provided in Figure 5 Figure 6

The GLM Procedure						
Tests of Hypotheses for Mixed Model Analysis of Variance						
Dependent Variable: time						
Source	DF	Type III SS	Mean Square	F Value	Pr > F	
person	3	151569	50523	28.74	< .0001	
color	3	59658	19886	11.31	0.0019	
Error	9.2423	16249	1758.075968			
Error: 0.9868*MS(person*color) + 0.0132*MS(Error)						

Source	DF	Type III SS	Mean Square	F Value	Pr > F	
person*color	9	15822	1757.976111	1.00	0.4569	
Error: MS(Error)	46	81216	1765.565217			

Figure 5: Proc GLM Results

color	time LSMEAN	95% Confidence Limits	
1	963.562500	939.850441	987.274559
2	938.312500	914.600441	962.024559
3	882.562500	858.850441	906.274559
4	949.687500	923.176612	976.198388

Least Squares Means for Effect color			
i	j	Difference Between Means	Simultaneous 95% Confidence Limits for LSMean(i)-LSMean(j)
1	2	25.250000	-21.026989 71.526989
1	3	81.000000	34.723011 127.276989
1	4	13.875000	-35.209159 62.959159
2	3	55.750000	9.473011 102.026989
2	4	-11.375000	-60.459159 37.709159
3	4	-67.125000	-116.209159 -18.040841

Figure 6: Proc GLM Results

If the method is specified to be type3, `proc mixed` produces identical results as `proc glm`.

```
proc mixed data=candle2 method=type3;
class person color;
model time=color;
random person person*color;
lsmeans color /cl pdiff adjust=Tukey;
run;
```

specifying `type=3` will yield same result as in "proc glm".

REML estimation is in general preferred over other methods of estimation.

 Download

 Print



Activity Details

Task: View this topic